

Key Homework 19

Math 277, Fall 2005

Ordinary Differential Equations

Dr. Goutziers

[-] Initializations

[> restart;

[-] 1a: Page 192

Since $y_1(t) = \cos(t)$ is a solution to

$$y'' - y' + y = \sin(t)$$

$y(t) = 5y_1(t) = 5\cos(t)$ is a solution to

$$y'' - y' + y = 5\sin(t)$$

[-] 2b: Page 192

Since $y_1(t) = \frac{\sin(2t)}{4}$ is a solution of

$$y'' + 2y' + 4y = \cos(2t)$$

and $y_2(t) = \frac{t}{4} - \frac{1}{8}$ is a solution of

$$y'' + 2y' + 4y = t$$

$y(t) = 2y_2(t) - 3y_1(t) = 2\left(\frac{t}{4} - \frac{1}{8}\right) - \frac{3\sin(2t)}{4} = \frac{t}{2} - \frac{3\sin(2t)}{4} - \frac{1}{4}$ is a solution to

$$y'' + 2y' + 4y = 2t - 3\cos(2t)$$

[-] 5

Since a particular solution of the nonhomogeneous equation is already provided, all we need to do is find the general solution of the corresponding homogeneous equation. First code the given information.

```
> deq:=diff(theta(t), t$2)-diff(theta(t), t)-2*theta(t)=1-2*t;  
theta_p:=t-1;
```

$$deq := \left(\frac{d^2}{dt^2} \theta(t) \right) - \left(\frac{d}{dt} \theta(t) \right) - 2 \theta(t) = 1 - 2t$$

$$theta_p := t - 1$$

Find the auxiliary equation of the corresponding homogeneous equation, compute the eigenvalues, and assemble the general solution of the homogeneous equation.

```
> deqh:=lhs(deq)=0;  
aux:=simplify(eval(subs(theta(t)=exp(r*t), deqh))/exp(r*t));  
evals:=solve(aux, r);
```

$$deqh := \left(\frac{d^2}{dt^2} \theta(t) \right) - \left(\frac{d}{dt} \theta(t) \right) - 2 \theta(t) = 0$$

$$aux := r^2 - r - 2 = 0$$

$$evals := 2, -1$$

```
> theta_gh:=add(c[k]*exp(evals[k]*t), k=1..2);
```

$$theta_{gh} := c_1 e^{(2t)} + c_2 e^{(-t)}$$

The general solution for the inhomogeneous equation is given by the general solution of the homogeneous equation plus a particular solution of the inhomogeneous equation.

```
> theta_g:=theta_gh+theta_p;
```

$$theta_g := c_1 e^{(2t)} + c_2 e^{(-t)} + t - 1$$

10: Page 192

Since the differential family of

```
> (exp(t)+t)^2=simplify(expand((exp(t)+t)^2));
```

$$(e^t + t)^2 = e^{(2t)} + 2e^t t + t^2$$

equals $\{e^{(2t)}, e^t, t e^t, 1, t, t^2\}$, it is finite. Hence, the method of undetermined coefficients can be applied.

17: Page 192

```
> deq:=diff(y(t), t$2)-y(t)=-11*t+1;
```

$$deq := \left(\frac{d^2}{dt^2} y(t) \right) - y(t) = -11t + 1$$

To find the general solution for this differential equation, we first compute the general solution of the corresponding homogeneous equation.

```
> deqh:=lhs(deq)=0;
```

```
aux:=simplify(eval(subs(y(t)=exp(r*t), deqh))/exp(r*t));
```

```
evals:=solve(aux, r);
```

$$deqh := \left(\frac{d^2}{dt^2} y(t) \right) - y(t) = 0$$

$$aux := r^2 - 1 = 0$$

$$evals := 1, -1$$

```
> ygh:=add(c[k]*exp(evals[k]*t), k=1..2);
```

$$ygh := c_1 e^t + c_2 e^{(-t)}$$

Next, we use the method of undetermined coefficients to generate a particular solution of the nonhomogeneous equation. The differential family of $-11t + 1$ is given by $\{1, t\}$, so we use a trial solution of the form

```
> ytry:=a[1]+a[2]*t;
```

$$ytry := a_1 + a_2 t$$

Substitute this expression into the inhomogeneous equation and solve for the parameters a_1 and a_2 .

```
> eq_p:=eval(subs(y(t)=ytry, deq));
```

```
val_p:=solve(identity(eq_p, t), {a[1], a[2]});
```

$$eq_p := -a_1 - a_2 t = -11t + 1$$

$$val_p := \{a_2 = 11, a_1 = -1\}$$

Construct the corresponding particular solution.

```
> yp:=subs(val_p, ytry);
```

$$yp := -1 + 11t$$

The general solution of the inhomogeneous equation is given by

```
> yg:=ygh+yp;
```

$$yg := c_1 e^t + c_2 e^{(-t)} - 1 + 11t$$

20: Page 192

```
> deq:=diff(y(theta), theta$2)+4*y(theta)=sin(theta)-cos(theta);
```

$$deq := \left(\frac{d^2}{d\theta^2} y(\theta) \right) + 4y(\theta) = \sin(\theta) - \cos(\theta)$$

To find the general solution for this differential equation, we first compute the general solution of the corresponding homogeneous equation.

```
> deqh:=lhs(deq)=0;
```

```
aux:=simplify(eval(subs(y(theta)=exp(r*theta), deqh))/exp(r*theta));
evals:=solve(aux, r);
```

$$\text{deqh} := \left(\frac{d^2}{d\theta^2} y(\theta) \right) + 4 y(\theta) = 0$$

$$\text{aux} := r^2 + 4 = 0$$

$$\text{evals} := 2I, -2I$$

```
> ygh:=c[1]*cos(2*theta)+c[2]*sin(2*theta);
```

$$ygh := c_1 \cos(2\theta) + c_2 \sin(2\theta)$$

Next, we use the method of undetermined coefficients to generate a particular solution of the nonhomogeneous equation. The differential family of $\sin(\theta) - \cos(\theta)$ is given by $\{\sin(\theta), \cos(\theta)\}$, so we use a trial solution of the form

```
> ytry:=a[1]*sin(theta)+a[2]*cos(theta);
```

$$ytry := a_1 \sin(\theta) + a_2 \cos(\theta)$$

Substitute this expression into the inhomogeneous equation and solve for the parameters a_1 and a_2 .

```
> eq_p:=eval(subs(y(theta)=ytry, deq));
```

```
val_p:=solve(identity(eq_p, theta), {a[1], a[2]});
```

$$eq_p := 3 a_1 \sin(\theta) + 3 a_2 \cos(\theta) = \sin(\theta) - \cos(\theta)$$

$$\text{val_p} := \left\{ a_2 = \frac{-1}{3}, a_1 = \frac{1}{3} \right\}$$

Construct the corresponding particular solution.

```
> yp:=subs(val_p, ytry);
```

$$yp := \frac{1}{3} \sin(\theta) - \frac{1}{3} \cos(\theta)$$

The general solution of the inhomogeneous equation is given by

```
> yg:=ygh+yp;
```

$$yg := c_1 \cos(2\theta) + c_2 \sin(2\theta) + \frac{1}{3} \sin(\theta) - \frac{1}{3} \cos(\theta)$$

```
>
```

25: Page 192

```
> deq:=diff(z(x), x$2)+z(x)=2*exp(-x);
```

$$\text{deq} := \left(\frac{d^2}{dx^2} z(x) \right) + z(x) = 2 e^{(-x)}$$

To find the general solution for this differential equation, we first compute the general solution of the corresponding homogeneous equation.

```
> deqh:=lhs(deq)=0;
```

```
aux:=simplify(eval(subs(z(x)=exp(r*x), deqh))/exp(r*x));
```

```
evals:=solve(aux, r);
```

$$\text{deqh} := \left(\frac{d^2}{dx^2} z(x) \right) + z(x) = 0$$

$$\text{aux} := r^2 + 1 = 0$$

$$\text{evals} := I, -I$$

```
> zgh:=c[1]*cos(x)+c[2]*sin(x);
```

$$zgh := c_1 \cos(x) + c_2 \sin(x)$$

Next, we use the method of undetermined coefficients to generate a particular solution of the nonhomogeneous equation. The differential family of $2e^{(-x)}$ is given by $\{e^{(-x)}\}$. Since $e^{(-x)}$ is not a solution to the homogeneous equation, we can use a trial solution of the form

```
> ztry:=C*exp(-x);
```

$$ztry := C e^{(-x)}$$

Substitute this expression into the inhomogeneous equation and solve for the parameter C .

```
> eq_p:=eval(subs(z(x)=ztry, deq));
```

```
val_p:=solve(identity(eq_p, x), {C});
```

$$eq_p := 2 C e^{(-x)} = 2 e^{(-x)}$$

$$val_p := \{C = 1\}$$

Construct the corresponding particular solution.

```
> zp:=subs(val_p, ztry);
```

$$zp := e^{(-x)}$$

The general solution of the inhomogeneous equation is given by

```
> zg:=zgh+zp;
```

$$zg := c_1 \cos(x) + c_2 \sin(x) + e^{(-x)}$$

Finally, we implement the initial conditions and solve for c_1 and c_2 .

```
> eq1:=eval(subs(x=0, zg)=0);
```

```
eq2:=eval(subs(x=0, diff(zg, x))=0);
```

```
val_c:=solve({eq1, eq2}, {c[1], c[2]});
```

$$eq1 := c_1 + 1 = 0$$

$$eq2 := -1 + c_2 = 0$$

$$val_c := \{c_2 = 1, c_1 = -1\}$$

The solution of the given initial value problem is

```
> sol:=subs(val_c, zg);
```

$$sol := -\cos(x) + \sin(x) + e^{(-x)}$$

```
>
```

28: Page 192

```
> deq:=diff(y(t), t$2)+diff(y(t), t)-12*y(t)=exp(t)+exp(2*t)-1;
```

$$deq := \left(\frac{d^2}{dt^2} y(t)\right) + \left(\frac{d}{dt} y(t)\right) - 12 y(t) = e^t + e^{(2t)} - 1$$

To find the general solution for this differential equation, we first compute the general solution of the corresponding homogeneous equation.

```
> deqh:=lhs(deq)=0;
```

```
aux:=simplify(eval(subs(y(t)=exp(r*t), deqh))/exp(r*t));
```

```
evals:=solve(aux, r);
```

$$deqh := \left(\frac{d^2}{dt^2} y(t)\right) + \left(\frac{d}{dt} y(t)\right) - 12 y(t) = 0$$

$$aux := r^2 + r - 12 = 0$$

$$evals := 3, -4$$

```
> ygh:=add(c[k]*exp(evals[k]*t), k=1..2);;
```

$$ygh := c_1 e^{(3t)} + c_2 e^{(-4t)}$$

Next, we use the method of undetermined coefficients to generate a particular solution of the nonhomogeneous equation. The differential family of $e^t + e^{(2t)} - 1$ is given by $\{e^t, e^{(2t)}, 1\}$, none of which is a solution to the homogeneous equation. Hence, we can use a trial solution of the form

```
> ytry:=a[1]*exp(t)+a[2]*exp(2*t)+a[3];
```

$$ytry := a_1 e^t + a_2 e^{(2t)} + a_3$$

Substitute this expression into the inhomogeneous equation and solve for the parameter C .

```
> eq_p:=eval(subs(y(t)=ytry, deq));
```

```
val_p:=solve(identity(eq_p, t), {a[1], a[2], a[3]});
```

$$eq_p := -10 a_1 e^t - 6 a_2 e^{(2t)} - 12 a_3 = e^t + e^{(2t)} - 1$$

$$val_p := \left\{a_3 = \frac{1}{12}, a_2 = \frac{-1}{6}, a_1 = \frac{-1}{10}\right\}$$

Construct the corresponding particular solution.

```
> yp:=subs(val_p, ytry);
```

$$yp := -\frac{1}{10}e^t - \frac{1}{6}e^{(2t)} + \frac{1}{12}$$

The general solution of the inhomogeneous equation is given by

> `yg:=ygh+yp;`

$$yg := c_1 e^{(3t)} + c_2 e^{(-4t)} - \frac{1}{10}e^t - \frac{1}{6}e^{(2t)} + \frac{1}{12}$$

Finally, we implement the initial conditions and solve for c_1 and c_2 .

> `eq1:=eval(subs(t=0, yg)=1);`
`eq2:=eval(subs(t=0, diff(yg, t))=3);`
`val_c:=solve({eq1, eq2}, {c[1], c[2]});`

$$eq1 := c_1 + c_2 - \frac{11}{60} = 1$$

$$eq2 := 3c_1 - 4c_2 - \frac{13}{30} = 3$$

$$val_c := \{c_2 = \frac{1}{60}, c_1 = \frac{7}{6}\}$$

The solution of the given initial value problem is

> `sol:=subs(val_c, yg);`

$$sol := \frac{7}{6}e^{(3t)} + \frac{1}{60}e^{(-4t)} - \frac{1}{10}e^t - \frac{1}{6}e^{(2t)} + \frac{1}{12}$$

>

35: Page 193

> `deq:=diff(y(t), t$2)-4*diff(y(t), t)+5*y(t)=exp(5*t)+t*sin(3*t)-cos(3*t);`

$$deq := \left(\frac{d^2}{dt^2}y(t)\right) - 4\left(\frac{d}{dt}y(t)\right) + 5y(t) = e^{(5t)} + t \sin(3t) - \cos(3t)$$

To find a particular solution of this differential equation, we first compute the eigenvalues of the corresponding homogeneous equation.

> `deqh:=lhs(deq)=0;`
`aux:=simplify(eval(subs(y(t)=exp(r*t), deqh))/exp(r*t));`
`evals:=solve(aux, r);`

$$deqh := \left(\frac{d^2}{dt^2}y(t)\right) - 4\left(\frac{d}{dt}y(t)\right) + 5y(t) = 0$$

$$aux := r^2 - 4r + 5 = 0$$

$$evals := 2 + I, 2 - I$$

Clearly, neither $e^{(5t)}$, nor $\sin(3t)$, nor $\cos(3t)$ is a solution to the homogeneous equation. So, we can use a trial solution of the form

$$a_1 e^{(5t)} + (b_0 + b_1 t) \sin(3t) + (c_0 + c_1 t) \cos(3t)$$

>

45: Page 193

45a

We solve two initial value problems. One for the time that the car is on the speed bump, that is for t on the interval

$\left[-\frac{\pi}{2V}, \frac{\pi}{2V}\right]$, the other for the time that the car has passed the speed bump, that is for t on the interval $\left[\frac{\pi}{2V}, \infty\right)$.

On $\left[-\frac{\pi}{2V}, \frac{\pi}{2V}\right]$, when the car is on the speed bump, we have the initial value problem

$$\left(\frac{d^2}{dt^2}y(t)\right) + y(t) = \cos(Vt)$$

$$y\left(-\frac{\pi}{2V}\right) = 0 \quad \text{and} \quad y'\left(-\frac{\pi}{2V}\right) = 0$$

Code the differential equation.

```
> deq:=diff(y(t), t$2)+y(t)=cos(V*t);
```

$$deq := \left(\frac{d^2}{dt^2} y(t)\right) + y(t) = \cos(Vt)$$

Find the general solution of the corresponding homogeneous equation.

```
> deqh:=lhs(deq)=0;
```

```
aux:=simplify(eval(subs(y(t)=exp(r*t), deqh))/exp(r*t));
```

```
evals:=solve(aux, r);
```

$$deqh := \left(\frac{d^2}{dt^2} y(t)\right) + y(t) = 0$$

$$aux := r^2 + 1 = 0$$

$$evals := I, -I$$

```
> ygh:=c[1]*cos(t)+c[2]*sin(t);
```

$$ygh := c_1 \cos(t) + c_2 \sin(t)$$

Create a particular solution of the inhomogeneous equation, assuming that $V \neq 1$.

```
> ytry:=a[1]*cos(V*t)+a[2]*sin(V*t);
```

```
eq_p:=eval(subs(y(t)=ytry, deq));
```

```
val_p:=solve(identity(eq_p, t), {a[1], a[2]});
```

$$ytry := a_1 \cos(Vt) + a_2 \sin(Vt)$$

$$eq_p := -a_1 \cos(Vt) V^2 - a_2 \sin(Vt) V^2 + a_1 \cos(Vt) + a_2 \sin(Vt) = \cos(Vt)$$

$$val_p := \{a_2 = 0, a_1 = -\frac{1}{V^2 - 1}\}$$

```
> yp:=subs(val_p, ytry);
```

$$yp := -\frac{\cos(Vt)}{V^2 - 1}$$

Create the general solution of the inhomogeneous equation.

```
> yg1:=ygh+yp;
```

$$yg1 := c_1 \cos(t) + c_2 \sin(t) - \frac{\cos(Vt)}{V^2 - 1}$$

Implement the initial conditions, and solve the initial value problem.

```
> eq1:=eval(subs(t=-Pi/(2*V), yg1)=0);
```

```
eq2:=eval(subs(t=-Pi/(2*V), diff(yg1, t))=0);
```

```
val_c1:=solve({eq1, eq2}, {c[1], c[2]});
```

$$eq1 := c_1 \cos\left(\frac{\pi}{2V}\right) - c_2 \sin\left(\frac{\pi}{2V}\right) = 0$$

$$eq2 := c_1 \sin\left(\frac{\pi}{2V}\right) + c_2 \cos\left(\frac{\pi}{2V}\right) - \frac{V}{V^2 - 1} = 0$$

$$val_c1 := \left\{ c_2 = \frac{\cos\left(\frac{\pi}{2V}\right) V}{V^2 - 1}, c_1 = \frac{V \sin\left(\frac{\pi}{2V}\right)}{V^2 - 1} \right\}$$

```
> soll:=subs(val_c1, yg1);
```

$$soll := \frac{V \sin\left(\frac{\pi}{2V}\right) \cos(t)}{V^2 - 1} + \frac{\cos\left(\frac{\pi}{2V}\right) V \sin(t)}{V^2 - 1} - \frac{\cos(Vt)}{V^2 - 1}$$

We use this result to compute the initial conditions for the initial value problem valid on the interval $\left[-\frac{\pi}{2V}, \infty\right)$. To do this, we

compute the values of $y(t)$ and $y'(t)$, for $t = \frac{\pi}{2V}$. We will denote these by yv and ypv respectively.

```
> yv:=simplify(eval(subs(t=Pi/(2*V), sol1)));
ypv:=simplify(eval(subs(t=Pi/(2*V), diff(sol1, t))));
```

$$yv := \frac{2V \sin\left(\frac{\pi}{2V}\right) \cos\left(\frac{\pi}{2V}\right)}{V^2 - 1}$$

$$ypv := \frac{2 \cos\left(\frac{\pi}{2V}\right)^2 V}{V^2 - 1}$$

Now, for t on the interval $[\frac{\pi}{2V}, \infty)$, when the car has passed the speed bump, we proceed by solving the initial value problem

$$\left(\frac{d^2}{dt^2} y(t)\right) + y(t) = 0$$

$$y\left(\frac{\pi}{2V}\right) = \frac{2V \sin\left(\frac{\pi}{2V}\right) \cos\left(\frac{\pi}{2V}\right)}{V^2 - 1} \quad \text{and} \quad y'\left(\frac{\pi}{2V}\right) = \frac{2 \cos\left(\frac{\pi}{2V}\right)^2 V}{V^2 - 1}$$

Recall the homogeneous differential equation.

```
> deqh;
```

$$\left(\frac{d^2}{dt^2} y(t)\right) + y(t) = 0$$

with general solution

```
> ygh;
```

$$c_1 \cos(t) + c_2 \sin(t)$$

Implement the initial conditions, and find the behavior of the car after it has passed the speed bump.

```
> eq3:=eval(subs(t=Pi/(2*V), ygh)=yv);
eq4:=eval(subs(t=Pi/(2*V), diff(ygh, t))=ypv);
val_c2:=solve({eq3, eq4}, {c[1], c[2]});
```

$$eq3 := c_1 \cos\left(\frac{\pi}{2V}\right) + c_2 \sin\left(\frac{\pi}{2V}\right) = \frac{2V \sin\left(\frac{\pi}{2V}\right) \cos\left(\frac{\pi}{2V}\right)}{V^2 - 1}$$

$$eq4 := -c_1 \sin\left(\frac{\pi}{2V}\right) + c_2 \cos\left(\frac{\pi}{2V}\right) = \frac{2 \cos\left(\frac{\pi}{2V}\right)^2 V}{V^2 - 1}$$

$$val_c2 := \left\{ c_2 = \frac{2 \cos\left(\frac{\pi}{2V}\right) V}{V^2 - 1}, c_1 = 0 \right\}$$

```
> sol2:=subs(val_c2, ygh);
```

$$sol2 := \frac{2 \cos\left(\frac{\pi}{2V}\right) V \sin(t)}{V^2 - 1}$$

Observe that this represents a sinusoidal oscillation with amplitude

$$\left| \frac{2V \cos\left(\frac{\pi}{2V}\right)}{V^2 - 1} \right|$$


45b

We plot the amplitude of the answer to Part(a) as a function of V .

```
> Ampl:=abs(coeff(sol2, sin(t)));
```

$$\text{Ampl} := 2 \left| \frac{\cos\left(\frac{\pi}{2V}\right) V}{V^2 - 1} \right|$$

```
> plot(Ampl, V=0.1..5, numpoints=1000, labels=[V, Amplitude]);
```



To find the V coordinate of the absolute maximum of the amplitude, we take the derivative and set it equal to zero.

```
> der:=diff(Ampl, V);
```

$$\text{der} := 2 \text{abs} \left(1, \frac{\cos\left(\frac{\pi}{2V}\right) V}{V^2 - 1} \right) \left(\frac{1}{2} \frac{\sin\left(\frac{\pi}{2V}\right) \pi}{V(V^2 - 1)} + \frac{\cos\left(\frac{\pi}{2V}\right)}{V^2 - 1} - \frac{2 \cos\left(\frac{\pi}{2V}\right) V^2}{(V^2 - 1)^2} \right)$$

In Maple, the function $\text{abs}(1, x)$ represents the derivative of the absolute value function $\text{abs}(x) = |x|$.

```
> Vmax:=fsolve(der=0, V=0.6..1);
```

$$V_{\text{max}} := 0.7313985404$$

```
>
```

The car and its inhabitants experience the most violent shakes if the speed bump is approached with a speed of .7313985404.